

DEFINE

MEASURE 'Products'[Unique Products] = DISTINCTCOUNT(Products[product_name])

MEASURE 'Transaction_Data'[Quantity Sold] = SUM(Transaction_Data[quantity])

MEASURE 'Transaction_Data'[Total Transactions] = COUNTROWS(Transaction_Data)

MEASURE 'Transaction_Data'[Weekend Transactions] = CALCULATE([Total Transactions],
'Calendar'[Weekend] = "Y")

MEASURE 'Transaction_Data'[% Weekend Transactions] = [Weekend Transactions] / [Total Transactions]

MEASURE 'Transaction_Data'[All Transactions] = CALCULATE([Total Transactions],
ALL(Transaction_Data))

MEASURE 'Transaction_Data'[Total Revenue] = SUMX(Transaction_Data,
Transaction_Data[quantity] * RELATED(Products[product_retail_price]))

MEASURE 'Transaction_Data'[Total Cost] = SUMX(Transaction_Data,
Transaction_Data[quantity] * RELATED(Products[product_cost]))

MEASURE 'Transaction_Data'[Total Profit] = [Total Revenue] - [Total Cost]

MEASURE 'Transaction_Data'[YTD Revenue] = CALCULATE([Total Revenue],
DATESYTD('Calendar'[date]))

MEASURE 'Transaction_Data'[60-Day Revenue] = CALCULATE([Total Revenue],
DATESINPERIOD('Calendar'[date], MAX('Calendar'[date]), -60, DAY))

MEASURE 'Transaction_Data'[Last Month Revenue] = CALCULATE([Total Revenue],
DATEADD('Calendar'[date], -1, MONTH))

MEASURE 'Transaction_Data'[Revenue Target] = [Last Month Revenue] * 1.05

MEASURE 'Transaction_Data'[Profit Margin] = [Total Profit] / [Total Revenue]

MEASURE 'Transaction_Data'[Last Month Transactions] = CALCULATE([Total Transactions],
DATEADD('Calendar'[date], -1, MONTH))

MEASURE 'Transaction_Data'[Last Month Profit] = CALCULATE([Total Profit],
DATEADD('Calendar'[date], -1, MONTH))

MEASURE 'Return_Data'[Quantity Returned] = SUM(Return_Data[quantity])

MEASURE 'Return_Data'[Total Returns] = COUNTROWS(Return_Data)

MEASURE 'Return_Data'[Return Rate] = [Quantity Returned] / [Quantity Sold]

MEASURE 'Return_Data'[All Returns] = CALCULATE([Total Returns], ALL(Return_Data))

MEASURE 'Return_Data'[Last Month Returns] = CALCULATE([Total Returns],
DATEADD('Calendar'[date],-1,MONTH))

EVALUATE

SUMMARIZECOLUMNS(

"Unique Products", 'Products'[Unique Products],

"Quantity Sold", 'Transaction_Data'[Quantity Sold],

"Total Transactions", 'Transaction_Data'[Total Transactions],

"Weekend Transactions", 'Transaction_Data'[Weekend Transactions],

"% Weekend Transactions", 'Transaction_Data'[% Weekend Transactions],

"All Transactions", 'Transaction_Data'[All Transactions],

"Total Revenue", 'Transaction_Data'[Total Revenue],

"Total Cost", 'Transaction_Data'[Total Cost],

"Total Profit", 'Transaction_Data'[Total Profit],

"YTD Revenue", 'Transaction_Data'[YTD Revenue],

"60-Day Revenue", 'Transaction_Data'[60-Day Revenue],

"Last Month Revenue", 'Transaction_Data'[Last Month Revenue],

"Revenue Target", 'Transaction_Data'[Revenue Target],

"Profit Margin", 'Transaction_Data'[Profit Margin],

"Last Month Transactions", 'Transaction_Data'[Last Month Transactions],

"Last Month Profit", 'Transaction_Data'[Last Month Profit],

"Quantity Returned", 'Return_Data'[Quantity Returned],

"Total Returns", 'Return_Data'[Total Returns],

"Return Rate", 'Return_Data'[Return Rate],

"All Returns", 'Return_Data'[All Returns],

"Last Month Returns", 'Return_Data'[Last Month Returns]

)